Newsgroup: it.comp.appl.access
Titolo: codici errori run-time

Ciao amichi dovo posso trovare tutti i codici (o meglio i più comuni) 
degli errori di Run-Time?
Riciao e grazie

 Riciao e grazie
Sulla guida in linea magari?
SunRex

 questa seconda funzione avrai i 4500 errori
Carlo in due sole funzioni riesci a compiere 5500 errori???
Ma come programmi???
gente incompetente...
MAssimiliano

 ha scritto nel messaggio 
news:kT0sd.37505$B06....@news.edisontel.com...

Eheeeee! Tutta invidia..
Carlo 


Questo codice è tratto dalla guida che ti invito caldamente a leggere...una 
miniera:
Esegui il codice e controlla la tabella errori appena creata.
Sub CreaTabellaErrori()
    Dim dbs As Database, tdf As TableDef, fld As Field
    Dim rst As Recordset, lngCodice As Long
    Const conErroreOggAppl = "Errore definito da applicazioni o da oggetti"
    ' Crea la tabella Errori in cui sono presenti i campi NumeroErrore e 
DescrizioneErrore.
    Set dbs = CurrentDb
    Set tdf = dbs.CreateTableDef("Errori")
    Set fld = tdf.CreateField("CodiceErrore", dbLong)
    tdf.Fields.Append fld
    Set fld = tdf.CreateField("StringaErrore", dbText, 255)
tdf.Fields.Append fld
    dbs.TableDefs.Append tdf
    ' Apre il set di record della tabella Errori.
    Set rst = dbs.OpenRecordset("Errori")
    ' Esegue un ciclo sui primi 1000 codici di errore di Visual Basic.
    For lngCodice = 1 To 1000
        On Error Resume Next
        ' Rileva ciascun errore.
        Err.Raise lngCodice
        DoCmd.Hourglass True
        ' Omette codici di errore che generano errori definiti da 
applicazioni o da oggetti.
        If Err.Description <> conErroreOggAppl Then
' Aggiunge ogni stringa e codice di errore alla tabella Errori.
            rst.AddNew
            rst!CodiceErrore = Err.Number
            rst!StringaErrore = Err.Description
            rst.Update
        End If
        ' Cancella oggetto Err.
        Err.Clear
    Next lngCodice
    ' Chiude il set di record.
    rst.Close
    DoCmd.Hourglass False
    MsgBox "Creata tabella Errori."
End Sub
Ciao, Carlo 


Nella prima routine che ti ho inviato puoi avere 1000 codici di errori, con 
questa seconda funzione avrai i 4500 errori di Jet ed tratta sempre dalla 
guida ma che nella versione originale aveva degli errori (sembra strano ma č 
cosě) ed io l'ho opportunamente corretta (sic!):
Se vuoi spingerti fino all'errore 31503 ti basta sostituire For lngCode = 0 
To 3500 con For lngCode = 0 To 32000
Function TabellaErroriAccessEJet() As Boolean
    Dim dbs As Database, tdf As TableDef, fld As Field
    Dim rst As Recordset, lngCode As Long
    Dim strErrAccess As String
    Const conErrOggApp = "Errore definito dall'applicazione o dall'oggetto"
    On Error GoTo Error_TabellaErroriAccessEJet
    ' Crea una tabella Errori con i campi NumeroErrore e DescrizioneErrore.
    Set dbs = CurrentDb
    Set tdf = dbs.CreateTableDef("ErroriAccessEJet")
    Set fld = tdf.CreateField("CodiceErrore", dbLong)
tdf.Fields.Append fld
    Set fld = tdf.CreateField("StringaErrore", dbMemo)
    tdf.Fields.Append fld
    dbs.TableDefs.Append tdf
    ' Apre il set di record della tabella Errori.
    Set rst = dbs.OpenRecordset("ErroriAccessEJet")
    ' Scorre i codici di errore in modo iterativo.
    For lngCode = 0 To 3500
        On Error Resume Next
        ' Trova ogni errore.
        strErrAccess = AccessError(lngCode)
        DoCmd.Hourglass True
        ' Salta i numeri di errore senza stringhe associate.
If strErrAccess <> "" Then
            ' Salta i codici che generano errori definiti dall'applicazione 
o dall'oggetto.
            If strErrAccess <> conErrOggApp Then
                ' Aggiunge ogni codice e stringa di errore alla tabella 
Errori.
                rst.AddNew
                rst!CodiceErrore = lngCode
                ' Accoda una stringa al campo memo.
                rst!StringaErrore.AppendChunk strErrAccess
                rst.Update
            End If
        End If
    Next lngCode
    ' Chiude il set di record.
rst.Close
    DoCmd.Hourglass False
    RefreshDatabaseWindow
    MsgBox "Tabella di errori Access e Jet creata."
    TabellaErroriAccessEJet = True
Exit_TabellaErroriAccessEJet:
    Exit Function
Error_TabellaErroriAccessEJet:
    MsgBox Err & ": " & Err.Description
    TabellaErroriAccessEJet = False
    Resume Exit_TabellaErroriAccessEJet
End Function
Ciao, Carlo